Title

Range-Energy Program for relativistic heavy ions in the region 1 < E < 3000 MeV/amu

Author

Salamon, M.

Publication Date

1980

Peer reviewed

UNIVERSITY OF CALIFORNIA

A RANGE-ENERGY PROGRAM FOR RELATIVISTIC HEAVY IONS IN THE REGION 1 < E < 3000 MeV/amu

MASTER

Michael H. Salamon

January 1980

Prepared for the U.S. Department of Energy under Contract W-7405-ENG-48

I. Introduction

Before the advent of relativistic heavy ion accelerators (1971 for LBL's Bevalac), the only practical need for accurate range-energy relations for relativistic heavy ions was restricted to the cosmic ray community. Since 1971, however, the need has increased in proportion to the expanded use of relativistic heavy ions in a wide variety of applications. At the present time, the most widely used range-energy tables are based on Bethe's expression for stopping power dE/dx, valid in the regime $Z_1\alpha/\beta << 1$, where $\alpha = e^2/\hbar c$, $\beta = v/c$, v = velocity of heavy ion projectile, and Z_1 = the actual charge of the heavy ion in the medium in which it is stopping. For $Z_1 = 1$ projectiles these tables are adequate for all but the lowest energies. As Z_1 increases, however, not only do we part from Bethe's regime of validity, but also additional, higher order (in Z_1) correction terms become significant, necessitating their inclusion in any stopping power expression whose accuracy is required to be $\lesssim 1\%$. To our knowledge, there are no published range-energy tables or computer programs which incorporate all these theoretically required corrections for the high Z_1, β regime, although the need for such tables is obvious.

This report describes a computer program (developed for analysis of Bevalac data) that calculates stopping power, range, or energy, and which includes the necessary corrections for accuracy in the high Z_1, β regime, as discussed in the theoretical and review papers of S.P. Ahlen.[1,2] The program, given in the appendix, is also available on PSS at LBL's Computer Center.

The following section discusses the expression for stopping power used in the range-energy program. The subsequent section explains how to use the code.

II. Stopping Power and Range

The most widely used relativistic stopping power formula, based on Bethe's first Born approximation quantum mechanical treatment,[3] is our "uncorrected" expression

$$\frac{dE}{dx} = \frac{4\pi N Z_2 Z_1{}^2 e^4}{mv^2} \left[\ln \frac{2mv^2}{I} + \ln\left(\frac{1}{1-\beta^2}\right) - \beta^2 - \frac{C}{Z^2} - \frac{\delta}{2}\right] \qquad (1)$$

where Z_2 = atomic number of the medium through which the heavy ion projectile passes, N = medium density (atoms/cm^3), m = electron mass, e = electron charge, I = logarithmic mean ionization potential of the medium (usually a parameter), and C/Z_2 and $\delta/2$ are the well known corrections accounting for the inner shell electron effect and the density effect (discussed below).

Bethe's formula being strictly proportional to $Z_1{}^2$ is due to two independent approximations. Traditionally, the problem of energy loss is facilitated by arbitrarily considering two types of collisions between the heavy ion projectile and the medium's electrons: close (hard) and distant (soft). In close (hard) collisions, the electron is assumed to be free (atomic binding forces are neglected). Bethe's treatment of close collisions uses the Mott scattering cross-section with only those terms in the expansion proportional to $Z_1{}^2$. The distant (soft) collision energy transfers are calculated in the dipole approximation, also giving a $Z_1{}^2$ dependence. Neither of these approximations is adequate as $Z_1{}^2$ increases. The exact Mott cross-section includes higher order Z_1 terms which should not be ignored (at $\beta = 0.9$ in Al, the correction to dE/dx is ≈20% for $Z_1 = 92$, ≈3% for $Z_1 = 26$). The use of the dipole approximation ignores gradients which contribute a Z_1^3 correction to stopping power which becomes significant at lower velocities.

Also, Bethe's use of the first Born approximation restricts the validity of the result to $Z_1\alpha/\beta \ll 1$. If $Z_1\alpha/\beta$ cannot be restricted to being $\ll 1$ (or $\gg 1$, in which case Bohr's classical treatment of energy loss[4,5] is applicable) then Bloch's expression,[6] valid for general $Z_1\alpha/\beta$, must be used. (Bloch's result came from an attempt to reconcile Bethe's quantum mechanical, $Z_1\alpha/\beta \ll 1$ result with Bohr's classical, $Z_1\alpha/\beta \gg 1$ result; his expression (at least non-relativistically) approaches both Bethe's and Bohr's in the appropriate limits.)

In addition, the equilibrium charge state Z_1 of the heavy ion projectile is generally not equal to its atomic number Z_0 because of electron pickup by the projectile.

There are thus three major corrections to the Bethe expression (eq. 1): the correction to the close collision stopping power due to higher order Z_1 Mott terms, the correction to the distant collision stopping power due to higher multipole interaction terms, and a term which converts the Bethe formula into the Bloch formula.

A. The Bloch Correction

Bloch's stopping power expression[6]

$$\frac{dE}{dx} = k\left[\ln\frac{2mv^2}{I} + \ln\left(\frac{1}{1-\beta^2}\right) - \beta^2 + \psi(1) - \mathrm{Re}\,\psi\left(1 + \frac{iZ_1\alpha}{\beta}\right) - \frac{C}{Z_2} - \frac{\delta}{2}\right] \quad (2)$$

$$k \equiv \frac{4\pi N Z_2 Z_1^2 e^4}{mv^2} \quad (3)$$

differs from Bethe's by $B = k[\psi(1) - \mathrm{Re}\psi(1 + \frac{iZ_1\alpha}{\beta})]$, where $\psi(z)$ is the digamma function. Using the expression[7]

$\mathrm{Re}\psi(1 + iy) = \psi(1) + y^2 \sum_{n=1}^{\infty} \frac{1}{n} (\frac{1}{n^2+y^2})$, $-\infty < y < \infty$, and since for $y << m$, $\frac{1}{m} (\frac{1}{m^2+y^2}) \approx \frac{1}{m^3}$, we get

$$B/k \approx -y^2[\sum_{n=1}^{m} \frac{1}{n} \{\frac{1}{n^2+y^2} - \frac{1}{n^2}\} + \zeta(3)] \ , \qquad (4)$$

where $\zeta(3) = 1.20206$ ($\zeta(z)$ is the Riemann-Zeta Function), and $y \equiv Z_1\alpha/\beta$. This is the expression used in the computer code for the Bloch term, with $m \sim 10\ y$. (An alternative expression for $\mathrm{Re}\psi(1 + iy)$ is $[1 + \psi(1) - \frac{1}{1+y^2} + \sum_{n=1}^{\infty} (-1)^{n+1} \{\zeta(2n + 1) - 1\} y^{2n}]$, $|y| < 2$, so $B/k = \sum_{n=1}^{\infty} (-1)^n \zeta(2n + 1) y^{2n}$, $|y| < 2$. For $|y| < 1$, the first term dominates, providing a Z_1^4 correction term to stopping power, $-1.202(\frac{Z_1\alpha}{\beta})$ k. Recent low energy stopping power measurements have demonstrated the importance of accuracy of this term in the regime $y \leq 1$[8] (where neither the Bethe nor the Bohr expressions apply))

B. The Mott Correction

Ahlen[1] has utilized the Mott cross-section expansions of Curr[9] and of Bartlett and Watson[10] to achieve an expression for the close collision stopping power containing terms up to Z_1^7. The resulting correction added to the Bethe expression (eq. 1) is $\frac{1}{2} kG(Z_1,\beta)$, where in our program

$$\begin{aligned} G(Z_1,\beta) \approx\ & (Z_1\alpha\beta)\ [1.725 + 0.52\pi \cos\chi] \\ & + (Z_1\alpha)^2\ (3.246 - 0.451\ \beta^2) \\ & + (Z_1\alpha)^3\ (1.522\ \beta + 0.987/\beta) \\ & + (Z_1\alpha)^4\ (4.569 - 0.494\ \beta^2 - 2.696/\beta^2) \\ & + (Z_1\alpha)^5\ (1.254\ \beta + 0.222/\beta - 1.170/\beta^3), \end{aligned}$$

where $\cos\chi \equiv \mathrm{Re}\{\frac{\Gamma(\frac{1}{2}-iy)\Gamma(1+iy)}{\Gamma(\frac{1}{2}+iy)\Gamma(1-iy)}\}$, $y \equiv \frac{Z_1\alpha}{\beta}$, and is tabulated in Ref. 1.

Based on estimates of the deviation by the Mott expansions used by Ahlen[1] from the exact Mott cross-section, Ahlen estimates the resulting fractional error in dE/dx to be $\frac{1}{6}\left(\frac{Z_1\alpha}{\beta}\right)^9$. The computer code thus ceases to include the Mott correction term when the associated error is greater than the fractional increases in dE/dx caused by $G(Z_1,\beta)$, i.e., when

$$k|G(Z_1,\beta)|/dE/dx \le \frac{1}{3}\left|\frac{Z_1\alpha}{\beta}\right|^9$$

C. Z_1^3 Low Velocity Correction

In going beyond the dipole approximation by including the next higher interaction multipole moment in the calculation of distant collision energy loss, Jackson and McCarthy,[11] following the work of Ashley, Ritchie and Brandt,[12] express the fractional correction to total energy loss as the universal function $Z_1F(V)/Z_2^{\frac{1}{2}}$, $V = 137\ \beta\gamma/Z_2^{\frac{1}{2}}$, where $V^2F(V)$ is a slowly varying function of V. Lindhard[13] has found that a low velocity close collision polarization correction also exists whose magnitude is comparable to the distant collision correction, giving a total fractional correction of $2Z_1F(V)/Z_2^{\frac{1}{2}}$. This result is corroborated both by Andersen et al.'s[8] stopping power measurements and by Heckman and Lindstrom's[14] measurement of range differences in emulsion for slow positive and negative pions. A very recent calculation of the Z_1^3 correction has been performed[21] by repeating Bethe's original treatment in the second order Born approximation. The result, ultimately semi-classical due to the unavoidable use of a cutoff parameter, corroborates the use of the Jackson-McCarthy plus Lindhard correction terms as implemented in this program.

In the computer code, $F(V) = 0.45V^{-2.5}$ for $V > 4$, while $V < 4$ interpolation between tabulated values is performed. Shell correction uncertainties limit the reliability of this calculation to $\beta > 0.07$; in the next subsection even stricter limits are imposed which restrict the use of our stopping power formula to $\beta > 0.13$, or $E > 8$ MeV/amu.

D. Inner Shell Correction

When the projectile's decreasing velocity becomes comparable to the velocity of the medium's inner shell electrons, certain assumptions implicit in the stopping power derivations fail (see Fano[15] for a detailed discussion) requiring the inclusion of an inner shell correction term C/Z_2. Experimental measurements of stopping at high energies designed to determine the single free parameter I (logarithmic mean ionization potential of the medium) have in the past assured the shell corrections to vanish as $\eta \equiv \beta\gamma \to \infty$. This has been shown to be incorrect for high Z_2 materials,[15] implying that the quantity I_{adj} rather than I is being measured, where

$$\ln I_{adj} + C_{adj}/Z_2 \equiv \ln I + C/Z_2 \ , \qquad (6)$$

with $C_{adj} \to 0$ as $\eta \to \infty$, and with all energy dependence resting in C_{adj}. Our code uses the fitting parameters of Barkas and Berger[16] to the asymptotic form of Walske[17] for C_{adj}:

$$C_{adj}(I_{adj},\eta) = (0.422377\ \eta^{-2} + 0.0304043\ \eta^{-4} - 0.00038106\ \eta^{-6})10^{-6}I_{adj}^{\ 3}$$
$$+ (3.858019\ \eta^{-2} - 0.1667989\ \eta^{-4} + 0.00157955\ \eta^{-6})10^{-9}I_{adj}^{\ 3}$$

where I_{adj} is in units of eV. This expression is not valid for $\beta < 0.13$.

This limitation restricts the use of our stopping power computer code (but not the range or energy code) to $E \geq 8$ MeV/amu.

E. Density Effect

The $\delta/2$ term in eq. 1 represents the reduction in stopping power caused by a decrease in electric field strength due to long range polarization in a dense medium, i.e., the density effect. First considered by Fermi,[18] it has been considered in detail by Sternheimer. Our computer code for δ is based on Sternheimer and Peierls[19] generalized formula

$$\delta = 2\ell n\beta\gamma - 2\ell n(I/\hbar\omega_p) - 1 + a(y_1 - 2\ell n\beta\gamma)^m; \quad y_0 < 2\ell n\beta\gamma < y_1$$

$$\delta = 2\ell n\beta\gamma - 2\ell n(I/\hbar\omega_p) - 1; \quad 2\ell n\beta\gamma > y_1$$

$$\delta = 0 \quad ; \quad 2\ell n\beta\gamma < y_0$$

where y_0, y_1, a, m are parameters determined by properties of the medium (density and mean ionization potential I) and ω_p is the medium's plasma frequency. The code avoids unnecessary computation by not calculating δ for $\gamma < 1.8$.

F. Effective Charge Z_1

The projectile charge Z_1 is the actual charge of the projectile, not its atomic number Z_0. For projectile velocities much larger than $Z_0\alpha c$, $Z_1 = Z_0$, but as the projectile slows it begins to pick up and retain electrons in its inner shells, reducing its effective charge. Pierce and Blann,[20] using differentiated range measurements of a number of low energy heavy ions in a variety of materials have established a universal functional form for Z_1 in terms of the atomic number Z_0, viz.,

$$Z_1 = Z_0[1 - \exp(-130\beta/Z_0^{2/3})] \qquad (9)$$

This is the value for Z_1 used by our programs. For Fe, e.g., $Z_1 = 25.5$ at ≈ 40 MeV/amu.

G. Range Below 8 MeV/amu

Since the shell correction formula is invalid for $E < 8$ MeV/amu, our relativistic dE/dx expression can not be used below this energy. Range calculations, however, require dE/dx down to $E > 0$ MeV/amu, since $R(E) = \int_0^E dE'/(\frac{dE'}{dx})$. For the region $0 < E \leq 8$ MeV/amu, we use the expression for range given by Barkas and Berger,[16]

$$R_{BB}(\beta) = \frac{M_1}{Z_1^2}\,[\lambda(\beta) + B_{Z_1}(\beta)]\ , \qquad (10)$$

where M_1 = mass of projectile in proton masses (m_p) and

$$\ell n\lambda(\beta) = \ell n\,\frac{A_2}{Z_2} + \sum_{n=0}^{2}\sum_{m=0}^{2} a_{mn}[\ell n I_{adj}]^m\,[\ell n\{(\gamma-1)m_p\}]^n \qquad (11)$$

is a least-squares analysis fit to low energy proton range data, and $B_{Z_1}(\beta)$ is an empirical range extension (based on light ion emulsion data) to account for electron pickup.

H. Multiple Coulomb Scattering

The projected range t (penetration depth) will be smaller than R, the actual projectile pathlength, due to multiple Coulomb scattering which causes a fractional difference[15,2]

$$\frac{\langle R-t\rangle}{R} \approx Z_2\,\frac{m}{M_1}\,\langle \ell \rangle\ , \qquad (12)$$

where $\langle \ell \rangle$ lies between 0.3 and 0.6. This difference, being only 2% for the worst case of protons on lead, usually may be ignored for heavy ion projectiles. The ranges calculated in this program are actual pathlengths R, and not projected ranges t.

I. Summary

The final expression for stopping power used in the program is

$$\frac{dE}{dx} = k[(\ln \frac{2mv^2}{I_{adj}} - \frac{C_{adj}}{Z_2})(1 + \frac{2Z_1F(V)}{Z_2^{\frac{1}{2}}}) + \ln(\frac{1}{1-\beta^2}) - \beta^2 + \frac{G}{2} + \frac{B}{k} - \frac{\delta}{2}]$$

where the various functions Z_1, δ, G, C_{adj}, etc., are given in the above equations. The range R is calculated by integration over a tabulated set of energy values $\{E_i\}$, where

$$R_i = R(E_i) = R(E_{i-1}) + \int_{E_{i-1}}^{E_i} dE/(dE/dx)$$

where the integral is evaluated by the 4-point Gauss-Legendre method. After the $\{R_i, E_i\}$ table is established, further calls to either the range or energy function involve only linear interpolation within the table. This is discussed in more detail in the next section. For $E_i < 8$ MeV/amu, the Barkas and Berger ranges are evaluated.

III. Use of the Code

A. Argument List

The argument lists for the functions RDEDX, RRANGE, RNERGY are identical, except that RNERGY's first argument is R(range) while RDEDX's and RRANGE's first argument is E(energy). The arguments, their meanings and units, are listed:

1. E: energy of the heavy ion projectile in MeV/amu. The maximum value is 3000 MeV/amu, in conformance with Bevalac energy limitations.

2. R: range, or actual pathlength of the heavy ion projectile in the medium, in units of g/cm^2.

3. Z1: this second argument is the atomic number of the projectile. The actual charge Z_1 is calculated internally.

4. A1: The mass, in amu, of the nuclear projectile.

5. Z2, A2, IADJ: these next three arguments characterize the medium in which the projectile is stopping. Z2 and A2 are the medium's charge and mass, and IADJ is the adjusted logarithmic mean ionization potential of the medium (in units of eV), and can be found tabulated for various materials,[2,15] or may be calculated from the formulae[16]

$$I_{adj} = \begin{cases} 12Z_2 + 7 \text{ eV}, & Z_2 \le 13 \\ 9.76Z_2 + 58.8Z_2^{-0.19} \text{ eV}, & Z_2 > 13. \end{cases}$$

For mixed media (several atomic species) Bragg's rule[15] is invoked (i.e. each atomic species contributes independently to the projectile slowing) to give

$$\langle A_2\rangle = \sum_i N_i (A_2)_i / N = \rho / [\sum_i \rho_i / (A_2)_i]$$

$$\langle Z_2\rangle = \sum_i N_i (Z_2)_i / N = \langle A_2\rangle \sum_i \rho_i \left(\frac{Z_2}{A_2}\right)_i / \rho$$

$$\langle \ell n I_{adj}\rangle = \sum_i N_i (Z_2)_i (\ell n I_{adj})_i / \sum_i N_i (Z_2)_i$$

$$= \frac{\langle A_2\rangle}{\rho\langle Z_2\rangle} \sum_i \left(\frac{Z_2}{A_2}\right)_i \rho_i (\ell n I_{adj})_i$$

where $N = \sum_i N_i$, N_i = # atoms of species i/cm^3, and $\rho = \sum_i \rho_i$, ρ_i density of species i in g/cm^3. (IADJ is a REAL variable.)

6. I1, I2, I3: these parameters are either 0 or 1, and turn "off" or "on" the higher order Z_1 corrections terms; i.e., the Mott, Bloch, and low velocity $Z_1{}^3$ corrections are respectively turned off if I1, I2, or I3 is set to zero. This allows the magnitude of the individual corrections to be determined. Normally, I1 = I2 = I3 = 1.

7. RHO, IGAS, ETA: these parameters are required for calculation of the density effect. If the stopping medium is a gas, IGAS = 1; otherwise, IGAS = 0. RHO is the density in g/cm^3 of the medium. For a gas, RHO is the gas density at 0° Centigrade, I atm pressure, and ETA is the factor which when multiplied by RHO gives the actual gas density. (This extra parameter is required by Sternheimer's empirical evaluation of the denisty effect in gases.)

B. Range-Energy Table Formation

The range-energy tables for a given projectile species differ for different stopping media. To facilitate stack calculations, i.e., particle slowing in multi-layered media, the range-energy tables calculated for a given {Z2, A2, IADJ, RHO, IGAS, ETA} are maintained in memory if a new medium is being entered by the projectile. The first call to RRANGE or RNERGY generates a new range-energy table (after which linear interpolation is used in subsequent calls) for that medium. Should the original medium be re-entered by the projectile, the correct range-energy tables are consulted by the program. Up to 20 simultaneous tables may be stored. If, however, Z_1, A_1, I1, I2, or I3 are changed, all previous range-energy tables are lost. Thus, to reduce calculation time, {Z_1, A_1, I1, I2, I3} should be changed as infrequently as possible.

References

1. S.P. Ahlen, Phys. Rev. A17, 1236 (1978).
2. S.P. Ahlen, Rev. Mod. Phys. (in press); also LBL Report No. 9353, Lawrence Berkeley Laboratory.
3. H. Bethe, Ann. Physik 5, 325 (1930); Z. Physik 76, 293 (1932).
4. N. Bohr, Phil. Mag. 25, 10 (1913); 30, 581 (1915).
5. N. Bohr, Kgl. Danske Videnskab. Selskab. Mat.-Fys. Medd. 18, No. 8 (1948).
6. F. Bloch, Ann. Phys. (Liepzig) 16, 285 (1933); Z. Physik 81, 363 (1933).
7. M. Abromowitz and I.A. Stegun, Handbook of Mathematical Functions, NBS Appl. Math Series 55 (1970).
8. H.H. Andersen, J.F. Bak, H. Knudson, and B.R. Nielsen, Phys. Rev. A16, 1929 (1977).
9. R.M. Curr, Proc. Phys. Soc. (London) A68, 156 (1955).
10. J.H. Bartlett and R.E. Watson, Proc. Am. Acad. Arts Sci. 74, 53 (1940).
11. J.D. Jackson and R.L. McCarthy, Phys. Rev. B6, 4131 (1972).
12. J.C. Ashley, R.H. Ritchie, and W. Brandt, Phys. Rev. B5, 2393 (1972).
13. J. Lindhard, Nucl. Instr. & Meth. 132, 1 (1976).
14. H.H. Heckman and P.J. Lindstrom, Phys. Rev. Lett. 22, 871 (1969).
15. U. Fano, Ann. Rev. Nucl. Sci. 13, 1 (1963).
16. W.H. Barkas and M.J. Berger, in Studies of Penetration of Charged Particles in Matter, Natl. Acad. Sci. Publ. 1133 (1964)
17. M.C. Walske, Phys. Rev. 88, 1283 (1952); 101, 940 (1956).

18. E. Fermi, Phys. Rev. 57, 485 (1940).

19. R.M. Sternheimer and R.F. Peierls, Phys. Rev. B3, 3681 (1971).

20. T.E. Pierce and M. Blann, Phys. Rev. 173, 390 (1968).

21. S.H. Morgan, Jr. and C.C. Sung, Phys. Rev. A20, 818 (1979).

RELATIVISTIC RANGE-ENERGY PROGRAMS

INTRODUCTION

THESE PROGRAMS ARE AVAILABLE AT LBL COMPUTER CENTER BY ACCESSING FROM PSS STORAGE. TO OBTAIN PROGRAM CODES PLUS THIS INTRODUCTION, EXECUTE THE CONTROL CARD

```
LIBCOPY,EXTUSER,YOURFIL,INTRO,RDEDX,RRANGE,RNERGY.
```

OR TO OBTAIN ONLY THE PROGRAM CODES, EXECUTE

```
LIBCOPY,EXTUSER,YOURFIL,RDEDX,RRANGE,RNERGY.
```

THIS CODE IS DESCRIBED IN DETAIL IN AN LBL REPORT ISSUED IN JANUARY, 1980. PLEASE ADDRESS ANY QUESTIONS OR CRITICISMS TO ITS AUTHOR,

MICHAEL SALAMON
345 BIRGE HALL, EXT 2-1958
PHYSICS DEPT, UCB

```
      FUNCTION RDEDX(E1,Z0,A1,Z2,A2,IADJ,I1,I2,I3,RHO,IGAS,ETAD)
C
C  THIS ROUTINE CALCULATES DE/DX USING THE BETHE EQUATION
C  WITH 3 CORRECTION TERMS, THE MOTT, BLOCH, AND LOW VELOCITY Z**3
C  TERMS. (SEE S.P. AHLEN, PRA17,1236(1978)). ANY OR ALL OF THESE
C  CORRECTIONS CAN BE INCLUDED OR IGNORED, SPECIFIED BY THE
C  INDICES I1(MOTT), I2(BLOCH), I3(LOW VELOCITY Z**3).  A ZERO INPUT
C  FOR A GIVEN PARAMETER ELIMINATES THAT PARTICULAR TERM IN THE
C  DE/DX CALCULATION.
C
C  THE SHELL CORRECTIONS ARE TAKEN FROM BARKAS AND BERGER,
C  PUBLICATION 1133 OF THE NATL ACAD SCI.  THE LOW VELOCITY Z**3 CORRECTION
C  IS DERIVED FROM A FIGURE IN MCCARTHY AND JACKSON, PHYS REV B6, 4131 (1972).
C  THE MOTT, BLOCH CORRECTIONS ARE FROM AHLEN'S PREVIOUSLY
C  REFERENCED PAPER.
C
C  UNITS OF E1=MEV/AMU
C  UNITS OF IADJ=EV.  IADJ IS A REAL VARIABLE.
C  UNITS OF RETURNED DE/DX=(MEV/AMU)/(G/CM2)
C
C  F1=STANDARD DE/DX FRONT FACTOR
C  F2=STANDARD BETHE NONRELATIVISTIC TERM WITH SHELL CORRECTIONS
C  F3=BLOCH CORRECTION TERM
C  F4=LOW VELOCITY Z**3 NONRELATIVISTIC CORRECTION FACTOR
C  F5=MOTT CORRECTION TERM (Z**3 TO Z**7)
C  F6=STANDARD BETHE RELATIVISTIC TERM
C
C  RHO=DENSITY OF MATERIAL, G/CM**3 (FOR A GAS, GIVE STANDARD DENSITY)
C  IGAS=0 IF CONDENSED PHASE, 1 IF GAS.
C  ETAD=FOR GAS, DENSITY RELATIVE TO STANDARD (1 ATM, 0 DEG CENT). MUST BE >0.
C
      COMMON/FLOOK/F1,F2,F3,F4,F5,F6,Z1
      REAL IADJ
      DIMENSION VA(4),V2FVA(4),Z1ABA(14),COSXA(14)
      DATA Z1ABA,COSXA/0.0,0.05,0.1,0.15,0.20,0.30,0.4,0.5,0.6,
     C  0.8,1.0,1.2,1.5,2.0,1.000,0.9905,0.9631,0.9208,0.8680,
     C  0.7478,0.6303,0.5290,0.4471,0.3323,0.2610,
     C  0.2145,0.1696,0.1261/
      DATA VA,V2FVA/1.,2.,3.,4.,0.33,0.30,0.26,0.23/
      PI=3.14159265
```

```
      ALP4A=1./137.03604
      G=1.+E1/931.5016
      DELT=DELTA(G,Z2,A2,IADJ,RHO,IGAS,ETAO)
      BSQ=1.-1./G**2
      B=SQRT(BSQ)
      Z1=Z0*(1.-EXP(-130.*B/Z0**(2./3.)))
      ETA=B*G
      EMASS=0.5110034E+06
      F1=0.3070722*Z1**2*Z2/(BSQ*A2)
      ETAM2=1./ETA**2
      CADJ=1.0E-06*IADJ**2*ETAM2*(0.422377+ETAM2*(0.0304043-ETAM2*
     1   0.00038106))+1.0E-09*IADJ**3*ETAM2*(3.858019+ETAM2*(-0.1667989
     2   +ETAM2*0.00157955))
      F2=ALOG(2.*EMASS*BSQ/IADJ)-CADJ/Z2
      F6=2.*ALOG(G)-BSQ
      F3=0.0
      F4=1.0
      F5=0.0
C
C  IF I2=0, DO NOT CALCULATE BLOCH CORRECTION.
      IF(I2.EQ.0)GO TO 60
      Y=Z1*ALPHA/B
      Y2=Y**2
      MSUM=INT(5.*Y)+1
      SUMR=0.
      DO 90 N=1,MSUM
      FN=FLOAT(N)
      FN2=FN**2
90    SUMR=SUMR+(1./(FN2+Y2)-1./FN2)/FN
      F3=-Y2*(1.202+SUMR)
C
C  IF I3=0, DO NOT CALCULATE LOW VELOCITY CORRECTION.
60    IF(I3.EQ.0)GO TO 50
      V=ETA/(ALPHA*SQRT(Z2))
      IF(V.GE.4.)GO TO 25
      DO 10 I=1,3
      IF(V.GE.VA(I+1))GO TO 10
      V2FV=V2FVA(I)+(V-VA(I))*(V2FVA(I+1)-V2FVA(I))
      GO TO 30
10    CONTINUE
25    V2FV=0.45/SQRT(V)
30    F4=1.+2.*Z1*V2FV/(V**2*SQRT(Z2))
C
C  IF I1=0, DO NOT CALCULATE MOTT CORRECTION.
50    IF(I1.EQ.0)GO TO 70
      Z1A=Z1*ALPHA
      Z1AB=ABS(Z1A/B)
      COSX=0.
      DO 40 I=1,13
      IF(Z1AB.GE.Z1ABA(I+1))GO TO 40
      COSX=COSXA(I)+(Z1AB-Z1ABA(I))*(COSXA(I+1)-COSXA(I))/
     C        (Z1ABA(I+1)-Z1ABA(I))
40    CONTINUE
      F5=0.5*Z1A*(B*(1.725+0.52*PI*COSX)+Z1A*(3.246-0.451*BSQ
     1   +Z1A*(1.522*3+0.987/B+Z1A*(4.569-0.494*BSQ-2.696/BSQ
     2    +Z1A*(1.254*B+0.222/B-1.170/BSQ/B)))))
      IF(Z1AB.LE.100.*ALPHA)GO TO 70
      IF((Z1AB**9/6.).LT.ABS(F5/(F2*F4+F3+F5+F6-DELT/2.)))GO TO 70
      F5=0.
70    RDEDX=F1*(F2*F4+F3+F5+F6-DELT/2.)/A1
      RETURN
      END
```

```
      FUNCTION DELTA(G,Z2,A2,FIADJ,RHO,IGAS,ETA)
C   THIS FUNCTION IS USED BY RDEDX.
C
C   THIS CORRECTION FOR THE DENSITY EFFECT IS BASED ON STERNHEIMER AND
C   PEIERLS, PHYS REV B3, 3681 (1971).
C   SET IGAS= 0 OR 1,  ETA > 0 REQUIRED.
C   RHO IS DENSITY IN G/CM**3.  FOR A GAS, GIVE RHO AT T=0 DEGREES
C   CENTIGRADE, 1 ATM PRESSURE, AND THE FACTOR ETA WHICH GIVES THE
C   ACTUAL GAS DENSITY UPON MULTIPLICATION BY RHO.
      IF(G.GE.1.8)GO TO 10
      DELTA=0.
      RETURN
10    PLASMA=28.8*SQRT(RHO*Z2/A2)
      CBAR=2.*ALOG(FIADJ/PLASMA)+1.0
      B=SQRT(1.-1./G**2)
      Y=2.*ALOG(B*G)+IGAS*ALOG(ETA)
      IF(IGAS.EQ.1)GO TO 100
      IF(FIADJ.GE.100.)GO TO 20
      Y1=9.212
      IF(CBAR.GE.3.681)GO TO 11
      Y0=0.9212
      GO TO 200
11    Y0=1.502*CBAR-4.606
      GO TO 200
20    Y1=13.82
      IF(CBAR.GE.5.215)GO TO 21
      Y0=0.9212
      GO TO 200
21    Y0=1.502*CBAR-6.909
      GO TO 200
100   IF(CBAR.GE.12.25)GO TO 110
      Y1=18.42
      IF(CBAR.LT.12.25)Y0=9.212
      IF(CBAR.LT.11.5)Y0=8.751
      IF(CBAR.LT.11.0)Y0=8.291
      IF(CBAR.LT.10.5)Y0=7.830
      IF(CBAR.LT.10.0)Y0=7.370
      GO TO 200
110   Y1=23.03
      IF(CBAR.GE.13.804)GO TO 120
      Y0=9.212
      GO TO 200
120   Y0=1.502*CBAR-11.52
200   A=(CBAR-Y0)/(Y1-Y0)**3
      IF(Y.GT.Y0)GO TO 210
      DELTA=0.
      RETURN
210   IF(Y.GE.Y1)GO TO 220
      DELTA=Y-CBAR+A*(Y1-Y)**3
      RETURN
220   DELTA=Y-CBAR
      RETURN
      END
      FUNCTION RRANGE(E,Z1,A1,Z2,A2,IADJ,I1,I2,I3,RHO,IGAS,ETAD)
C   THIS ROUTINE RETURNS RANGE IN G/CM2 FOR INPUTS E IN MEV/AMU;
C   Z1,Z2 CHARGE OF ION AND MEDIUM; A1,A2 IN AMU; IADJ IN EV.
C   MAXIMUM E= 3 GEV/AMU
C   THE RANGE IS CALCULATED USING THE RDEDX FUNCTION WHICH RETURNS
C   THE BETHE DE/DX WITH MOTT, BLOCH, AND LOW VEL Z**3 CORRECTIONS
C   ADDED. ANY OF THESE CORRECTIONS MAY BE DELETED BY SPECIFYING
C   I1(MOTT), I2(BLOCH), I3(LOW VELOCITY Z**3) EQUAL TO ZERO.
C
```

```
C  ALONG WITH THE PARTICULAR VALUE OF RANGE REQUESTED, AN ARRAY
C  IS RETJRNED IN COMMON/RCMMN/, TRANGE, WHICH GIVES RRANGE(E) FOR
C  E TABULATED IN COMMON ARRAY TENERG.
C
C  AFTER THE FIRST CALL, IF PARAMETERS OTHER THAN E DO NOT CHANGE,
C  LINEAR INTERPOLATION IS USED TO CALCULATE RRANGE USING THE
C  PREVIOJSLY CALCULATED TRANGE TABLE.
C  THE RANGE TABLES FOR UP TO 20 DIFFERENT MEDIA (WITH SAME ION
C  SPECIE) CAN BE CALCULATED AND STORED FOR SUBSEQUENT CALLS.  THIS
C  FACILITATES STACK CALCULATIONS. IF EXTERNAL JSE OF THE RANGE-
C  ENERGY TABLES IS REQUIRED, THE EXTERNAL PROGRAM MUST INCLUDE
C  COMMON/PARAC/ IN ORDER TO CORRELATE THE PARTICULAR IADJ, Z2, A2
C  WITH THE APPROPRIATE TRANGE VECTOR.
C
      REAL IADJ,IADJA(20)
      COMMON/PARAC/Z1P,A1P,I1P,I2P,I3P,NIADJ,IADJA,Z2A(20),A2A(20)
     1        ,RHOA(20),IGASA(20),ETADA(20)
      COMMON/RCMMN/MIADJ,TENERG(138),TRANGE(138,20)
      DATA TENERG/1.,2.,3.,4.,5.,6.,7.,8.,9.,10.,12.,14.,16.,18.,
     C  20.,22.,24.,26.,28.,30.,32.,34.,36.,38.,40.,42.,44.,46.,
     C  48.,50.,55.,60.,65.,70.,75.,80.,85.,90.,95.,100.,105.,
     C  110.,115.,120.,125.,130.,135.,140.,145.,150.,155.,160.,
     C  165.,170.,175.,180.,185.,190.,195.,200.,210.,220.,230.,
     C  240.,250.,260.,270.,280.,290.,300.,310.,320.,330.,340.,
     C  350.,360.,370.,380.,390.,400.,410.,420.,430.,440.,450.,
     C  460.,470.,480.,490.,500.,510.,520.,530.,540.,550.,560.,
     C  570.,580.,590.,600.,610.,620.,630.,640.,650.,660.,670.,
     C  680.,690.,700.,710.,720.,730.,740.,750.,760.,770.,780.,
     C  790.,800.,820.,840.,860.,880.,900.,920.,940.,960.,980.,
     C  1000.,1200.,1400.,1600.,1800.,2000.,2400.,2800.,3200./
      IF((Z1.EQ.Z1P).AND.(A1.EQ.A1P).AND.
     C  (I1.EQ.I1P).AND.
     C  (I2.EQ.I2P).AND.(I3.EQ.I3P))GO TO 451
C  PARAMETERS HAVE CHANGED.  GENERATE THE FIRST OF A NEW SET OF TABLES.
      NIADJ=1
      MIADJ=1
      Z1P=Z1
      A1P=A1
      I1P=I1
      I2P=I2
      I3P=I3
      IADJA(1)=IADJ
      Z2A(1)=Z2
      A2A(1)=A2
      RHOA(1)=RHO
      IGASA(1)=IGAS
      ETADA(1)=ETAD
      GO TO 450
C  PARAMETERS SAME, BUT IS THIS A NEW IADJ,Z2,A2,RHO,IGAS,ETAD?
451   DO 420 I=1,NIADJ
      MIADJ=I
C  IF NOT, GO DIRECTLY TO INTERPOLATION SECTION.
      IF((IADJ.EQ.IADJA(I)).AND.(Z2.EQ.Z2A(I)).AND.
     1  (A2.EQ.A2A(I)).AND.(RHO.EQ.RHOA(I)).AND.(IGAS.EQ.IGASA(I)).
     2   AND.(ETAD.EQ.ETADA(I)))GO TO 100
420   CONTINUE
C  IF SO, GENERATE NEXT IADJ TABLE.
      NIADJ=NIADJ+1
      IF(NIADJ.GE.20)NIADJ=20
      MIADJ=NIADJ
      IADJA(MIADJ)=IADJ
      Z2A(MIADJ)=Z2
```

```
      A2A(MIADJ)=A2
      RHOA(MIADJ)=RHO
      IGASA(MIADJ)=IGAS
      ETADA(MIADJ)=ETAD
C  FOR E < 9 MEV/AMU, WE USE AN EMPIRICAL FIT TO PROTON RANGE DATA
C  GIVEN BY BARKAS AND BERGER.
C  THE USE OF Z1EFF IN RDEDX AUTOMATICALLY INCORPORATES RANGE
C  EXTENSION DUE TO ELECTRON PICKUP IN THE RANGE CALCULATION FOR E>8
C  MEV/AMU.  FOR E<9 MEV/AMU, WHERE THE ANALYTIC FORM FOR THE SHELL CORRECTION
C  FAILS AND AN EMPIRICAL FIT TO PROTON RANGE DATA IS USED, A RANGE EXTENSION
C  GIVEN BY BARKAS + BERGER IN THIS REGION IS USED.
450   DO 30 I=1,8
30    TRANGE(I,MIADJ)=(A1*(931.5016/938.213)/Z1**2)*PRNGLO(TENERG(I),
     1      IADJ,Z2,A2)+BZ(TENERG(I),Z1,A1,Z2,A2,IADJ)
C  THIS SECTION INTEGRATES 1./DE/DX TO OBTAIN RANGE. THE METHOD IS
C  THAT OF GAUSS-LEGENDRE WITH 4 SAMPLE POINTS FOR INTEGRATION, THE
C  RANGES BEING INTEGRATD FROM TABULATED POINT TO TABULATED POINT.
      DO 300 I=9,138
      DE2=(TENERG(I)-TENERG(I-1))/2.
C
      DEDX1=RDEDX((TENERG(I-1)+1.33998104*DE2),Z1,A1,Z2,A2,IADJ,
     C     I1,I2,I3,RHO,IGAS,ETAD)
      DEDX2=RDEDX((TENERG(I-1)+1.86113631*DE2),Z1,A1,Z2,A2,IADJ,
     C     I1,I2,I3,RHO,IGAS,ETAD)
      DEDX3=RDEDX((TENERG(I-1)+0.13886369*DE2),Z1,A1,Z2,A2,IADJ,
     C     I1,I2,I3,RHO,IGAS,ETAD)
      DEDX4=RDEDX((TENERG(I-1)+0.66001869*DE2),Z1,A1,Z2,A2,IADJ,
     C     I1,I2,I3,RHO,IGAS,ETAD)
      DR=DE2*(0.65214515/DEDX1 + 0.34785485/DEDX2
     C        +0.34785485/DEDX3 + 0.65214515/DEDX4)
300   TRANGE(I,MIADJ)=TRANGE(I-1,MIADJ)+DR
100   IF(E.GT.TENERG(1))GO TO 60
      RRANGE=E*TRANGE(1,MIADJ)/TENERG(1)
      RETURN
60    DO 70 I=2,138
      IF(E.GT.TENERG(I))GO TO 70
      RRANGE=TRANGE(I-1,MIADJ)+(E-TENERG(I-1))
     C     *(TRANGE(I,MIADJ)-TRANGE(I-1,MIADJ))
     C     /(TENERG(I)-TENERG(I-1))
      GO TO 200
70    CONTINUE
200   CONTINUE
      RETURN
      END
      FUNCTION BZ(E,Z1,A1,Z2,A2,IADJ)
C  THIS FUNCTION IS USED BY RRANGE.
C
C  THIS ROUTINE RETURNS THE TOTAL RANGE EXTENSION TO RANGE
C  AS GIVEN BY BARKAS AND BERGER. THE A1/Z1**2 FACTOR IS
C  INCLUDED WITHIN THE ROUTINE, SO THE RETURNED VALUE IS TO BE
C  ADDED DIRECTLY TO THE CONSTANT CHARGE STATE RANGE.
C
C  UNITS OF BZ=G/CM2, E=MEV/AMU.
C
      REAL IADJ
      G=1.+E/931.5016
      B=SQRT(1.-1./G**2)
      IF(B.GE.(2.*Z1/137.))GO TO 10
      BZ=A1*(931.481/938.259)*(48.0+5.8*IADJ**(5./8.))*
     C     (A2/Z2)*1.0E-05*Z1**(-1./3.)*B
      GO TO 20
10    BZ=(A1*(931.481/938.259))*(7.0+0.85*IADJ**(5./8.))*
```

```
      C     (A2/Z2)*1.0E-06*Z1**(2./3.)
20     CONTINUE
       RETURN
       END
       FUNCTION PRNGLO(E,IADJ,Z2,A2)
C   THIS FUNCTION IS USED BY RRANGE.
C
C   THIS ROUTINE (PROTON-RANGE-LOW) CALCULATES PROTON RANGES
C   FOR 1<E<9 MEV/AMU. THIS USES THE EMPIRICAL RANGE EXPRESSION
C   BY BARKAS AND BERGER . RETURNED RANGE IS IN G/CM2. IT MUST
C   BE MULTIPLIED BY A1*(931.5016/938.213)/Z1**2 FOR CHARGES HIGHER
C   THAN 1. A1 IS IN AMU.
C
       REAL IADJ
       CR=938.213/931.5016
       ELN=ALOG(E*CR)
       ALN=ALOG(IADJ)
       RLN=ALOG(A2/Z2)-7.5265E-01+2.5398*ELN-2.4598E-01*ELN**2
      C  +7.3736E-02*ALN-3.1200E-01*ELN*ALN
      C  +1.1548E-01*ALN*ELN**2+4.0556E-02*ALN**2
      C  +1.8664E-02*ALN**2*ELN-9.9661E-03*ALN**2*ELN**2
       PRNGLO=EXP(RLN)/1.0E+03
       RETURN
       END
       FUNCTION RNERGY(R,Z1,A1,Z2,A2,IADJ,I1,I2,I3,RHO,IGAS,ETAD)
C   THIS ROUTINE IS THE COUNTERPART OF RRANGE, THE CORRECTED RANGE.
C   RRANGE IS CALLED, PRODUCING A RANGE-ENERGY TABLE. RNERGY IS
C   DETERMINED BY LINEAR INTERPOLATION ON THIS TABLE.
C
C   UNITS= R=G/CM2
C          A1,A2=AMU
C          IADJ=EV
C          RNERGY=MEV/AMU
C
       REAL IADJ, IADJA(20)
       COMMON/PARAC/Z1P,A1P,I1P,I2P,I3P,NIADJ,IADJA,Z2A(20),A2A(20)
      1       ,RHOA(20),IGASA(20),ETADA(20)
       COMMON/RCMMN/MIADJ,TENERG(138),TRANGE(138,20)
C   PICK A VALUE OF E IN CALLING RRANGE
       E=600.
C   THIS CONSTRUCTS THE RANGE-ENERGY TABLE TO BE USED BELOW IF ONE
C   DOES NOT ALREADY EXIST.
       RR=RRANGE(E,Z1,A1,Z2,A2,IADJ,I1,I2,I3,RHO,IGAS,ETAD)
       IF(R.LE.TRANGE(1,MIADJ))GO TO 20
       DO 10 I=2,138
       K=I
       IF(R.LE.TRANGE(I,MIADJ))GO TO 30
10     CONTINUE
30     RNERGY=TENERG(K-1)+(R-TRANGE(K-1,MIADJ))*(TENERG(K)-TENERG(K-1))/
      C                 (TRANGE(K,MIADJ)-TRANGE(K-1,MIADJ))
       RETURN
20     RNERGY=TENERG(1)*R/TRANGE(1,MIADJ)
C   OBVIOUSLY, THIS FUNCTION IS INACCURATE BELOW E=1 MEV/AMU
       RETURN
       END
```